

Библиотека констант и функций, общих для всего домашнего задания.

DoB, DoP - начальные коэффициенты диффузии для бора и фосфора соответственно, см²/с

NsB, NsP - поверхностная концентрация бора и фосфора на кремнии, ат/см³

WB, WP - энергия активации диффузии бора и фосфора, эВ

K - константа Больцмана, (Дж/К)/(Дж/эВ) = эВ/К

fNB(x) - содержание бора, ат/см³ при сопротивлении x, Ом*см

fNP(x) - содержание фосфора, ат/см³ при сопротивлении x, Ом*см

fDB(T), fDP(T) - коэффициенты диффузии бора и фосфора при данной температуре, см²/с;

fQB(T,t), fQP(T,t) - мощность источника после загонки с температурой T временем t, 1/см²;

fCiB(T,x,t), fCiP(T,x,t) - распределение концентрации по глубине, ат/см³, бесконечный источник

fClimB(T,Q,x,t), fClimP(T,Q,x,t) - то же, ограниченный источник

Для всех функций: температура - в С, время - в часах, толщина и глубина в сантиметрах.

```
> SharedFuncs := module()  
> export DoB,DoP,NsB,NsP,WB,WP,K,fNB,fNP,fDB,fDP,fQB,fQP,  
>       fCiB,fCiP,fClimB,fClimP;  
> local locfDB,locfDP;  
>  
> DoB := 10.5;  
> DoP := 10.5;  
> NsB := 1e21;  
> NsP := 1e22;  
> WB := 3.66;  
> WP := 3.66;  
> K := 1.380662*1e-23 / (1.60219*1e-19);  
>  
> fNB := (r) -> 10^(16.3-1.23*log10(r));  
> fNP := (r) -> 10^(15.9-1.37*log10(r));  
>  
> fDB := (T) -> DoB * exp(-WB/K/(T+273));  
> fDP := (T) -> DoP * exp(-WP/K/(T+273));  
>  
> locfDB := (T,t) -> fDB(T)*t*3600;  
> locfDP := (T,t) -> fDP(T)*t*3600;  
>  
> fQB := (T,t) -> 2 * NsB * sqrt(locfDB(T,t)/Pi);  
> fQP := (T,t) -> 2 * NsP * sqrt(locfDP(T,t)/Pi);  
>  
> fCiB := (T,x,t) -> NsB * erfc(x/2/sqrt(locfDB(T,t)));  
> fCiP := (T,x,t) -> NsP * erfc(x/2/sqrt(locfDP(T,t)));  
>  
> fClimB := (T,Q,x,t) -> Q/sqrt(Pi*locfDB(T,t)) * exp(-x*x/(4*locfDB(T,t)));  
> fClimP := (T,Q,x,t) -> Q/sqrt(Pi*locfDP(T,t)) * exp(-x*x/(4*locfDP(T,t)));  
>  
> end module;  
>  
>  
> savelib(SharedFuncs);
```